

Polymodal な度数分布を正規分布へ分解する BASIC プログラムの検討

赤 嶺 達 郎¹⁾

Consideration of the BASIC Programs to Analyse the Polymodal Frequency Distribution into Normal Distributions.

TATSURO AKAMINE¹⁾

Abstract

A Maximum-Likelihood program was compared with a Least-Squares program and its variations. The algorithm for convergence was MARQUARDT's method according to AKAMINE (1984). Elements and eigenvalues · eigenvectors of the inverse of an Hessian matrix in the neighborhood of the solution were calculated for a quadratic approximation to estimate errors. A likelihood ratio test was used to estimate the confidence interval of the maximum eigenvector. The obtained results are summarized as follows:

- 1) The Maximum Likelihood method is the most adequate procedure for this problem.
- 2) The χ^2 minimum method is more adequate than the Least-Squares method for normal data, but the latter is more adequate than the former for abnormal data which have a few separate parts at the end of a distribution. These methods are easy to apply for a robust estimation.
- 3) Parameters are stable in the part where an obvious minimal value is recognized between neighboring distributions. On the other hand, the confidence intervals of the parameters are larger than for the parts where it is not recognized.

Polymodal な度数分布を正規分布へ分解する手法は年齢形質が得られていない生物種を扱う場合等に用いられる基本的な手法である。古くからさまざまな手法が開発されており、嶋津(1979)、松崎ほか(1983)に概略が紹介されている。これらの大部分は正規分布を1つずつ度数分布にあてはめていく方法で、最近ではBASICプログラム化されており、労力の軽減や精度の向上がはかられている。これらの中で解の客観性からみて最良の方法は非線型最小二乗法であり、嶋津(1979)は大型計算機のサブルーチンを用い、赤嶺(1982b, 1984)は小型計算機用のBASICプログラムを作成した。

1985年1月14日受理 日本海区水産研究所業績A第425号

1) 〒951 新潟市水道町1丁目5939-22 日本海区水産研究所
(Japan Sea Regional Fisheries Research Laboratory, Suido-cho, Niigata 951, Japan)

しかし、観測値に理論度数曲線をあてはめる場合には、最尤法が最小二乗法や積率法よりも原理的にすぐれていることは FISHER が既に指摘している (安藤・門脇 1980). HASSELBLAD (1966) は大型計算機を用いてこの問題に最尤法を適用したが、これは大型計算機用のプログラムであるため一般の研究者にはあまり普及していない。

本報では MARQUARDT 法を用いた最小二乗法の BASIC プログラム (赤嶺 1984) を最尤法に書きかえ、応用例を使用して最小二乗法とその変形法 (以下、最小二乗法系手法と言う) と比較するとともに、パラメータの誤差推定および具体的な使用手順等についても検討した。

I. プログラムの説明

1 定式化

赤嶺 (1984) の最小二乗法 (MARQUARDT 法) のプログラムを最尤法に書きかえたので変数等はすべて赤嶺 (1984) と同一とした。ただし、パラメータの表示は α から θ に変更した。

与えられた度数分布を F , 階級の数 m , 階級の中 h , 階級値の最小値 a , 最大値 b とする。ただし、階級値は各階級の中央値である。分解する正規分布の数を n とおくと、求める式 f は

$$f = \sum_{i=1}^n K_i N(\mu_i, \sigma_i, x) \quad \text{---①}$$

$$N(\mu_i, \sigma_i, x) = \frac{1}{\sqrt{2\pi} \sigma_i} \exp \left\{ -\frac{(x - \mu_i)^2}{2\sigma_i^2} \right\}$$

となる (図 1)。ここで




図 1 変数の説明

Fig. 1. Illustration of variables.

$$S = \int_{-\infty}^{\infty} f dx = \sum_i K_i \quad \text{---②}$$

$$P_i = \frac{K_i}{S} \quad \text{---③}$$

$$g = \sum_{i=1}^n P_i N(\mu_i, \sigma_i, x) = \frac{f}{S} \quad \text{---④}$$

とおけば、 g は確率分布となる。 f において独立なパラメータ数は $3n$ であるが、 g では $3n-1$

である。

そこで尤度関数を次のように定義できる。

$$\log L = \sum_{x=x_1}^{x_T} \log g \quad \text{---⑤}$$

$$T = \sum_{x=a}^b F, \quad G = \frac{F}{T} \quad \text{---⑥}$$

ここで T は総個体数, G は相対度数分布である。

⑤式を最大にするパラメータの値を求めるのが最尤法であり, $n=1$ の場合の解は解析的に求まり $P=1$, μ =標本平均, σ^2 =標本分散(不偏分散ではない)となる。⑤式を使えば度数分布を作成せずにデータから直接計算できる。つまり度数分布の作り方に左右されない。

しかし, データ数が多くなると小型計算機で⑤式を使用するのは非効率であり, また最小二乗法等との比較のためからも⑤式を度数分布に適用してプログラムを作成する。したがって⑤式は

$$\log L = \sum_{x=a}^b F \log g \quad \text{---⑦}$$

となる。

HASSELBLAD (1966) は

$$\sum_i P_i = 1 \quad \text{---⑧} \quad \text{より}$$

$$P_n = 1 - \sum_{i=1}^{n-1} P_i \quad \text{---⑨}$$

として, ⑨式を④式に代入して解を求めた。赤嶺(1984)をこの方法に作りかえることは容易である。具体的には

$$\frac{\partial g}{\partial \mu_i} = N(\mu_i, \sigma_i, x) - N(\mu_n, \sigma_n, x) \quad \text{---⑩}$$

($i = 1 \sim n-1$)

に注意して P_n をのぞく $(3n-1)$ 個のパラメータについて MARQUARDT 法で解を求めればよい。赤嶺(1984)では HENSU (N3) に P_n の値がはいっているので, GAUSS の消去法において N3 を $N3-1$ に変更すればよい。プログラム例をプログラム2に示す。

ここでは最小二乗法との記述を統一するため, 以下のようにして MARQUARDT 法を適用した。

目的関数を Y とすると

$$\begin{aligned} Y &= -\log L = -\sum_x F \log g = -\sum F \log \frac{f}{S} \\ &= -\sum F (\log f - \log S) \\ &= -\sum F \log f + T \log S \end{aligned} \quad \text{---⑪}$$

となる。⑪式を最小にする $3n$ 個のパラメータについて MARQUARDT 法で解を求めればよい。パラメータを1つ増やしたため P_i は1つの値に収束するが, S が不定となるため K_i も不定となる。つまり解の近傍で誤差曲面は楕円面とならず楕円面が一方方向に無限にひきのばされた形となっている。 S の値は累積度数分布との比較から考えて最後に $S=T$ とした(プログラムでは120行で行なっている)。

2 MARQUARDT 法

HASSELBLAD (1966) は最急降下法 (steepest descent method) と NEWTON 法を用いて

解を求めた。しかし、HASSELBLAD の使用した最急降下法は本来の最急降下法ではなく、むしろ GAUSS-SEIDEL 法に近い手法である。したがって収束はかなり遅いと考えられる。また NEWTON 法では発散を防止するため、あらかじめ最急降下法で何回か反復計算した値を初期値として使用している。

最急降下法は安定であるが収束が遅く、NEWTON 法はこの逆の性質を持っている。両者を融合したのが MARQUARDT 法であり、アルゴリズムが簡単なわりによく働くので小型計算機で関数の極値を求める手法としては現在のところ最適な手法であると考えられる。

最急降下法は目的関数： Y ，パラメータ： $\theta = (\theta_1, \dots, \theta_n)$ とおくと

$$\Delta\theta = kg \tag{12}$$

$$g = \left(-\frac{\partial Y}{\partial \theta} \right)$$




図2 最急降下法
Fig. 2. Convergence of steepest descent method in the case of 2 linear parameters.

として θ を修正していく手法で、きざみ巾 k はそのつど修正する(図2)。 g を最急降下ベクトルと呼ぶ。 g はパラメータのスケールリングによって変化するので適切なスケールリングが必要である。また g は極小値の方向に向いているので極大値を求める場合には $-g$ を使用する。

NEWTON 法は

$$H\Delta\theta = g \tag{13}$$

$$H = \left(\frac{\partial^2 Y}{\partial \theta_i \partial \theta_j} \right)$$

として θ を修正していく手法である(図3)。 H をヘシアン行列と呼ぶ。NEWTON 法は線型に近い場合数回で収束するが、非線型性が強い場合振動・発散を起こす。なお 13式において連立方程式を各パラメータごとに順々に解いて反復する方法が GAUSS-SEIDEL 法の原理で収束は遅いが安定している。

MARQUARDT 法は 12式と 13式をくっつけて

$$(H + \lambda I) \Delta\theta = g \tag{14}$$

(I : 単位行列)




図3 NEWTON法と GAUSS-SEIDEL法

Fig. 3. Convergence of NEWTON's method and GAUSS-SEIDEL's method in the case of 2 linear parameters.

NEWTON's method: $P_0 \rightarrow$ the point of intersection of lines A and B.

GAUSS-SEIDEL's method: $P_0 \rightarrow P_1 \rightarrow P_2 \rightarrow P_3 \rightarrow \dots$

$$A: \frac{\partial Y}{\partial \theta_1} = 0, \quad B: \frac{\partial Y}{\partial \theta_2} = 0$$

としたもので、 λ が大きいときには最急降下法に近づき、 λ が0に近いときにはNEWTON法に近づく。極小値を求める場合、手順として最初 λ を大きくとっておき、 $\Delta Y < 0$ のときには $\lambda^{NEW} = \lambda^{OLD} / \nu$ で小さくしていき、 $\Delta Y \geq 0$ のときには $\lambda^{NEW} = \lambda^{OLD} \cdot \nu$ で大きくしてやり直す。 λ が連続して大きくなる場合は解に到達したか停留点にはいっていると判断して計算を打ち切る。一般的には $\nu = 2$ であるが、大型計算機では非線型性の程度によって λ を調整する修正 MARQUARDT法が開発されている(中川・小柳 1982)。

3 最尤法

目的関数: Y , パラメータ: θ とおくと

$$Y = \sum_x y(f(\theta)) \quad \text{---(15)}$$

$$\frac{\partial Y}{\partial \theta} = \sum_x \frac{\partial y}{\partial f} \frac{\partial f}{\partial \theta} \quad \text{---(16)}$$

$$\frac{\partial^2 Y}{\partial \theta_i \partial \theta_j} = \sum_x \left(\frac{\partial^2 y}{\partial f^2} \frac{\partial f}{\partial \theta_i} \frac{\partial f}{\partial \theta_j} + \frac{\partial y}{\partial f} \frac{\partial^2 f}{\partial \theta_i \partial \theta_j} \right) \quad \text{---(17)}$$

である。⑰式の第2項は第1項に比較して寄与が小さいので最小二乗法では省略している。最尤法でも第2項を省略しても無事収束したが反復回数が多くなった。第2項を計算する分だけ1回の計算時間は長くなるが、ここでは第2項を省略しないでプログラムを作成した。

$$y = -F \log \frac{f}{S} \quad \text{より}$$

$$\begin{cases} \frac{\partial y}{\partial f} = -\frac{F}{f} + \frac{F}{S} \frac{\partial S}{\partial f} \\ \frac{\partial^2 y}{\partial f^2} = -\frac{F}{f^2} - \frac{F}{S^2} \left(\frac{\partial S}{\partial f}\right)^2 + \frac{F}{S} \frac{\partial^2 S}{\partial f^2} \end{cases}$$

これを

$$\frac{\partial^2 S}{\partial \theta_i \partial \theta_j} = \frac{\partial^2 S}{\partial f^2} \frac{\partial f}{\partial \theta_i} \frac{\partial f}{\partial \theta_j} + \frac{\partial S}{\partial f} \frac{\partial^2 f}{\partial \theta_i \partial \theta_j}$$

に注意して (16), (17) 式に代入すると

$$\frac{\partial Y}{\partial \theta_i} = -\sum \frac{F}{f} \frac{\partial f}{\partial \theta_i} + \frac{T}{S} \frac{\partial S}{\partial \theta_i} \quad \text{---(18)}$$

$$\begin{aligned} \frac{\partial^2 Y}{\partial \theta_i \partial \theta_j} &= -\sum F \left(-\frac{1}{f^2} \frac{\partial f}{\partial \theta_i} \frac{\partial f}{\partial \theta_j} + \frac{1}{f} \frac{\partial^2 f}{\partial \theta_i \partial \theta_j} \right) \\ &\quad - \frac{T}{S^2} \frac{\partial S}{\partial \theta_i} \frac{\partial S}{\partial \theta_j} + \frac{T}{S} \frac{\partial^2 S}{\partial \theta_i \partial \theta_j} \end{aligned} \quad \text{---(19)}$$

を得る。ただしこの場合は (11) 式より直接に求めた方が簡単である。

このとき

$$\begin{cases} \frac{\partial f}{\partial K_i} = N(\mu_i, \sigma_i, x) \\ \frac{\partial f}{\partial \mu_i} = K_i N(\mu_i, \sigma_i, x) \frac{x - \mu_i}{\sigma_i^2} \\ \frac{\partial f}{\partial \sigma_i} = K_i N(\mu_i, \sigma_i, x) \frac{(x - \mu_i)^2 - \sigma_i^2}{\sigma_i^3} \end{cases} \quad \text{---(20)}$$

$\frac{\partial^2 f}{\partial \theta_i \partial \theta_j}$ は

$$\begin{cases} \frac{\partial^2 f}{\partial K_i \partial \mu_i} = N(\mu_i, \sigma_i, x) \frac{x - \mu_i}{\sigma_i^2} \\ \frac{\partial^2 f}{\partial K_i \partial \sigma_i} = N(\mu_i, \sigma_i, x) \frac{(x - \mu_i)^2 - \sigma_i^2}{\sigma_i^3} \\ \frac{\partial^2 f}{\partial \mu_i^2} = K_i N(\mu_i, \sigma_i, x) \frac{(x - \mu_i)^2 - \sigma_i^2}{\sigma_i^4} \\ \frac{\partial^2 f}{\partial \mu_i \partial \sigma_i} = K_i N(\mu_i, \sigma_i, x) \frac{(x - \mu_i)^3 - 3(x - \mu_i) \sigma_i^2}{\sigma_i^5} \\ \frac{\partial^2 f}{\partial \sigma_i^2} = K_i N(\mu_i, \sigma_i, x) \frac{(x - \mu_i)^4 - 5(x - \mu_i)^2 \sigma_i^2 + 2 \sigma_i^4}{\sigma_i^6} \end{cases} \quad \text{---(21)}$$

となっており、他はすべて 0。また

$$\begin{cases} \frac{\partial S}{\partial K_i} = 1 \\ \frac{\partial S}{\partial \mu_i} = \frac{\partial S}{\partial \sigma_i} = \frac{\partial^2 S}{\partial \theta_i \partial \theta_j} = 0 \end{cases} \quad \text{である。} \quad \text{---(22)}$$

スケーリングは

$$\begin{aligned} \theta' &= \frac{\theta}{a} && \text{としているので} \\ \frac{\partial Y}{\partial \theta'} &= \frac{\partial Y}{\partial \theta} \frac{\partial \theta}{\partial \theta'} && \text{より} \end{aligned}$$

$$\begin{cases} \frac{\partial Y}{\partial \theta'_i} = a_i \frac{\partial Y}{\partial \theta_i} \\ \frac{\partial^2 Y}{\partial \theta'_i \partial \theta'_j} = a_i a_j \frac{\partial^2 Y}{\partial \theta_i \partial \theta_j} \\ \Delta \theta_i = a_i \Delta \theta'_i \end{cases} \quad \text{---(23)}$$

a_i としてはその時の各パラメータの値を使用した (図 4).




図 4 パラメータのスケーリング

Fig. 4. Scaling of parameters.

g : steepest descent vector.

g' : steepest descent vector of scaling parameters.

なお連立方程式の解法には GAUSS の消去法を採用した。H は対称行列なので上三角部分だけで計算している (戸川 1971)。

4 最小二乗法手法

この問題について最小二乗法には二つの方法が考えられる。第 1 の方法は赤嶺 (1982 b, 1984) のように度数分布: F に f をあてはめる方法であり、第 2 の方法は相対度数分布: G に g をあてはめる方法である。第 1 の方法で求めた解を f_o, S_o とすると、 $S_o \neq T$ なので $g'_o = f_o/S_o, G=F/T$ より、 g'_o は第 2 の方法の解 g_o とは一致しない。つまり第 1 の方法において $S=T$ という制限をつけたもの (このため独立なパラメータ数は 1 つ減少する) が第 2 の方法と一致する。良いデータでは $S \approx T$ となるので両手法の差は小さいが、変形法等では大きくなる可能性がある。第 2 の方法の方が一般的であるが、第 1 の方法の方があてはまりはよいと考えられる。ここでは両手法を採用し、第 1 の方法をプログラム 3、第 2 の方法をプログラム 4 として作成し検討した。

まず第 1 の方法について述べる。最小二乗法および比較検討したその変形法の目的関数とその微分式を以下に示す。

$$\begin{cases} y = (F-f)^2 \\ \frac{\partial y}{\partial f} = -2(F-f) \\ \frac{\partial^2 y}{\partial f^2} = 2 \end{cases} \quad \text{---24}$$

$$\begin{cases} y = \frac{(F-f)^2}{f} \\ \frac{\partial y}{\partial f} = -\frac{F^2 - f^2}{f^2} = 1 - \frac{F^2}{f^2} \\ \frac{\partial^2 y}{\partial f^2} = 2 \frac{F^2}{f^3} \end{cases} \quad \text{---25}$$

$$\begin{cases} y = \frac{(F-f)^2}{f^2} = \left(\frac{F}{f} - 1\right)^2 \\ \frac{\partial y}{\partial f} = -2 \frac{F(F-f)}{f^3} \\ \frac{\partial^2 y}{\partial f^2} = 2 \frac{F(3F-2f)}{f^4} \end{cases} \quad \text{---26}$$

$$\begin{cases} y = (\log F - \log f)^2 = \left(\log \frac{F}{f}\right)^2 \\ \frac{\partial y}{\partial f} = -2 \frac{\log F - \log f}{f} \\ \frac{\partial^2 y}{\partial f^2} = 2 \frac{1 + \log \frac{F}{f}}{f^2} \end{cases} \quad \text{---27}$$

$$\begin{cases} y = \left\{ \log(F+1) - \log(f+1) \right\}^2 \\ \frac{\partial y}{\partial f} = -2 \frac{\log(F+1) - \log(f+1)}{f+1} \\ \frac{\partial^2 y}{\partial f^2} = 2 \frac{1 + \log(F+1) - \log(f+1)}{(f+1)^2} \end{cases} \quad \text{---28}$$

②4式は普通の最小二乗法，②5式は χ^2 最小化法，②6式は F/f が1に近くなるようにした最小二乗法，②7式は対数変換した値についての最小二乗法，ただしこれでは $F=0$ の値は使えないので1を加えて修正したものが②8式である．これらの式を①5～①7式に代入することによって解を得ることができる（プログラム3）．

次に第2の方法について述べる． F, f を用いて $S=T$ という制限をつけてもよいが，まぎらわしいので G, g を用いた．まず F を G に直し，②4～②8式において F, f を G, g に書きかえればよい．後の処理はプログラム2と全く同様で，⑩式（プログラムでは960行）だけに注意し GAUSS の消去法において変数を $3n$ から $3n-1$ に変更するだけである．プログラム2からの変更部分をプログラム4に示した．

最小二乗法は度数分布の誤差 $\epsilon = F-f$ が正規分布に従うことを仮定して最尤法を適用したものである．Polymodalな度数分布ではこの仮定はあてはまらない．特に f が0に近い場合 $F=f+\epsilon < 0$ となる部分があって不合理である．したがってこの場合には便利な計算方法という意味しか持たない．

5 プログラムの説明

赤嶺(1984)の最小二乗法のプログラムを最尤法に作りかえたものを以下に示す．プログラム1は⑩式による方法で，プログラム2はそれを⑨式による方法への書きかえ，プログラム3, 4は最小二乗法系の方法(②4～②8式)への書きかえである．入力データ例は田中(1956)

で使用された キダイデータの異常値をカットしたものである。プログラムの概略を以下に示す。

40～ 140	メインルーチン
200～ 540	変数, データ, パラメータの初期値入力
600～ 700	Y_0 の計算
800～1190	H, g の計算
960～1040	$\frac{\partial f}{\partial \theta}, \frac{\partial^2 f}{\partial \theta_i \partial \theta_j}$ の計算 (スケーリング)
1090～1170	y の計算
1150～1170	$\frac{T}{S}, \frac{T}{S^2}$ の修正 (スケーリング)
1300～1650	$\Delta \theta = H^{-1}g$ の計算, $\Delta Y < 0$ の判定, λ および θ の修正
1440	(スケーリング)
1700～1840	$Y, f, K_i N(\mu_i, \sigma_i, x)$ の計算
2000～2180	GAUSS の消去法
2500～2590	出力
3000～	データ

$$\left(\begin{array}{l} n, a, m, h, \text{ 反復回数}, \lambda, \nu \\ F_a, F_{a+h}, F_{a+2h} \dots, F_b \\ K_1, \mu_1, \sigma_1, \dots, K_n, \mu_n, \sigma_n \end{array} \right)$$

Program 1. Program of the maximum likelihood method (object function : $L=F$
 (その1) ($\log f - \log S$), number of parameters= $3n$, DATA : the example of porgy)

```

10  REM
20  REM POLYMODAL 5 ( MAXIMUM LIKELIHOOD - MARQUARDT )
30  REM
40  REM MAIN
50  GOSUB *NYURYOKU
60  GOSUB *SHOKICHI
70  FOR KK1=1 TO NIT+1
80      GOSUB *SHUKEI
90      GOSUB *KEISAN
100 NEXT KK1
110 *OWARI
120 FOR I=1+2*NND TO N3 : HENSU(I)=HENSU(I)*SOWA/SOWA2 : NEXT I
130 GOSUB *SHUTU
140 END
200 *NYURYOKU
210 READ NND, MCM, NCL, CWD, NIT, LAMBDA, NU
220 PRINT "セイフ`ンフ` ノ カス" = ; NND
230 PRINT "サイショウ ノ カイクウチ=" ; MCM
240 PRINT "カイクウ ノ カス"   = ; NCL
250 PRINT "カイクウ ノ ハハ"   = ; CWD
260 PRINT "      ハンフ`クスウ  = ; NIT
270 PRINT "      LAMBDA      = ; LAMBDA
280 PRINT "      NU          = ; NU
290 PRINT
300 STOP
310 N3=3*NND
320 DIM F(NCL), X(NCL), GX(NCL), BIBUN(N3), ND(NND, NCL), KEISU2(N3)
330 DIM HENSU(N3), KEISU(N3, N3), TEISU(N3), ZOBUN(N3), HENSU2(N3)
340 SOWA=0
350 FOR K=1 TO NCL
360     X(K)=MCM+(K-1)*CWD
370     READ F(K)
380     PRINT "F("; X(K); ")=" ; F(K)
390     SOWA=SOWA+F(K)
400 NEXT K
410 PRINT
420 PRINT "ソウワ=" ; SOWA
430 PRINT
440 STOP
450 SOWA2=0
460 FOR I=1 TO NND
470     S3=I : S2=S3+NND : S1=S2+NND
480     READ HENSU(S1), HENSU(S2), HENSU(S3)
490     PRINT "I=" ; I
500     PRINT "メンセキ=" ; HENSU(S1), "ヘイキン=" ; HENSU(S2), "フ`ンサン=" ; HENSU(S3)
510     SOWA2=SOWA2+HENSU(S1)
520 NEXT I
530 STOP
540 RETURN
600 *SHOKICHI
610 P9=.398942 : P8=-.5
620 FOR I=1 TO N3
630     HENSU2(I)=HENSU(I)
640 NEXT I
650 SOWA3=SOWA2
660 GOSUB *CLL2
670 LL1=LL2
680 PRINT "Y=" ; LL1
690 PRINT
700 RETURN
    
```

プログラム1 (その2)

```

800  *SHUKEI
810  REM ショキカ
820  FOR I=1 TO N3
830    TEISU(I)=0
840    FOR J=1 TO N3
850      KEISU(I,J)=0
860    NEXT J : NEXT I
870  REM ソウウ
880  FOR K=1 TO NCL
890    F9=F(K) : G1=GX(K) : F7=F9/G1
900    FOR J=1 TO NND
910      S3=1 : S2=S3+NND : S1=S2+NND
920      P1=HENSU(S1) : P2=X(K)-HENSU(S2) : P3=HENSU(S3)
930      P4=HENSU(S2)
940      P6=ND(I,K)
950      PP2=P2*P2 : PP3=P3*P3
960      BIBUN(S1)=P6
970      BIBUN(S2)=P6*P2/PP3*P4
980      BIBUN(S3)=P6*(PP2-PP3)/PP3
990      F8=F7*P6 : PPP2=PP2*PP2 : PPP3=PP3*PP3
1000     KEISU(S3,S3)=KEISU(S3,S3)-F8*(PPP2-5*PP2*PP3+2*PPP3)/PPP3
1010     KEISU(S3,S2)=KEISU(S3,S2)-F8*P2*(PP2-3*PP3)/PPP3*P4
1020     KEISU(S3,S1)=KEISU(S3,S1)-F8*(PP2-PP3)/PP3
1030     KEISU(S2,S2)=KEISU(S2,S2)-F8*(PP2-PP3)/PPP3*P4*P4
1040     KEISU(S2,S1)=KEISU(S2,S1)-F8*P2/PP3*P4
1050   NEXT J
1060   REM
1070   F6=F7/G1
1080   FOR I=1 TO N3
1090     TEISU(I)=TEISU(I)+F7*BIBUN(I)
1100     FOR J=I TO N3
1110       KEISU(I,J)=KEISU(I,J)+F6*BIBUN(I)*BIBUN(J)
1120     NEXT J : NEXT I : NEXT K
1130   SS1=SOWA/SOWA2 : SS2=SS1/SOWA2
1140   FOR I=1+2*NND TO N3
1150     TEISU(I)=TEISU(I)-SS1*HENSU(I)
1160     FOR J=I TO N3
1170       KEISU(I,J)=KEISU(I,J)-SS2*HENSU(I)*HENSU(J)
1180     NEXT J : NEXT I
1182   FOR I=2 TO N3-1
1183     KEISU2(I)=KEISU(I,I)
1184     KEISU(I,1)=TEISU(I)
1185   NEXT I
1186   FOR I=2 TO N3-1
1187     FOR J=I+1 TO N3
1188       KEISU(J,I)=KEISU(I,J)
1189   NEXT J : NEXT I
1190   RETURN

```

プログラム1 (その3)

```
1300 *KEISAN
1310 LAMBDA2=0
1320 K2=0
1330 *REP
1340 K2=K2+1
1350 IF K2>11 GOTO *OWARI
1360 PRINT
1370 PRINT "LAMBDA=" ;LAMBDA
1380 PRINT
1390 FOR I=1 TO N3
1400     KEISU(I, I)=KEISU(I, I)+LAMBDA-LAMBDA2
1410 NEXT I
1420 GOSUB *GAUSS
1430 FOR I=1 TO N3
1440     ZOBUN(I)=ZOBUN(I)*HENSU(I)
1450     HENSU2(I)=HENSU(I)+ZOBUN(I)
1460 NEXT I
1470 SOWA3=0
1480 FOR I=1+2*NND TO N3
1490     SOWA3=SOWA3+HENSU2(I)
1500 NEXT I
1510 GOSUB *CLL2
1520 IF LL2<LL1 THEN GOTO *PSKIP
1521 LAMBDA2=LAMBDA : LAMBDA=LAMBDA*NU
1522 FOR I=2 TO N3
1523     KEISU(I, I)=KEISU2(I)
1524     TEISU(I)=KEISU(I, 1)
1525 NEXT I
1526 FOR I=2 TO N3-1
1527     FOR J=I+1 TO N3
1528         KEISU(I, J)=KEISU(J, I)
1529 NEXT J : NEXT I
1530 GOTO *REP
1535 *PSKIP
1540 LAMBDA=LAMBDA/NU
1550 DELTA=LL2-LL1
1560 PRINT
1570 PRINT "DELTA=" ;DELTA
1580 PRINT
1590 LL1=LL2
1600 FOR I=1 TO N3
1610     HENSU(I)=HENSU2(I)
1620 NEXT I
1630 SOWA2=SOWA3
1640 GOSUB *SHUTU
1650 RETURN
```

プログラム1 (その4)

```

1700 *CLL2
1710 LL2=0
1720 FOR K=1 TO NCL
1730     F1=0
1740     FOR I=1 TO NND
1750         S3=I : S2=S3+NND : S1=S2+NND
1760         P1=HENSU2(S1) : P2=X(K)-HENSU2(S2) : P3=HENSU2(S3)
1770         ND(I,K)=P9*P1/P3*EXP(P8*P2*P2/P3/P3)
1780         F1=F1+ND(I,K)
1790     NEXT I
1800     LL2=LL2-F(K)*LOG(F1)
1810     GX(K)=F1
1820 NEXT K
1830 LL2=LL2+SOWA*LOG(SOWA3)
1840 RETURN
2000 *GAUSS
2010 REM センシン ショウキョ
2020 FOR I=1 TO N3-1
2030     FOR K=I+1 TO N3
2040         Q1=KEISU(I,K)/KEISU(I,I)
2050         TEISU(K)=TEISU(K)-Q1*TEISU(I)
2060         FOR J=K TO N3
2070             KEISU(K,J)=KEISU(K,J)-Q1*KEISU(I,J)
2080 NEXT J : NEXT K : NEXT I
2090 REM コウタイ タ`イニユウ
2100 ZOBUN(N3)=TEISU(N3)/KEISU(N3,N3)
2110 FOR I=N3-1 TO 1 STEP -1
2120     T1=TEISU(I)
2130     FOR J=I+1 TO N3
2140         T1=T1-ZOBUN(J)*KEISU(I,J)
2150     NEXT J
2160     ZOBUN(I)=T1/KEISU(I,I)
2170 NEXT I
2180 RETURN
2500 *SHUTU
2510 PRINT "Y=";LL1
2520 PRINT
2530 PRINT "ハンフ`クスウ=";KK1
2540 FOR I=1 TO NND
2550     PRINT "I=";I
2560     S3=I : S2=S3+NND : S1=S2+NND
2570     PRINT "メンセキ=";HENSU(S1), "ヘイキン=";HENSU(S2), "フ`ンサン=";HENSU(S3)
2580 NEXT I
2590 RETURN
3000 DATA 5, 7. 5, 24, 1, 20, 10000, 2
3010 DATA 7, 79, 509, 2240, 2341, 623, 476, 1230, 1439, 921, 448, 512, 719, 673
3020 DATA 445, 341, 310, 228, 168, 140, 114, 64, 22, 0
3030 DATA 5000, 11, 1, 4000, 15. 5, 1, 3000, 20, 1. 5, 1000, 24, 1. 5, 500, 27, 1. 5

```

Program 2. Parts of change from program 1 to the maximum likelihood method
 (object function : $L = F \log g$, $P_n = 1 - \sum_{i=1}^{n-1} P_i$, number of parameters = $3n - 1$)

```
55 FOR I=1+2*NND TO N3 : HENSU(I)=HENSU(I)/SOWA2 :NEXT I
120 FOR I=1+2*NND TO N3+1 : HENSU(I)=HENSU(I)*SOWA : NEXT I
960 BIBUN(S1)=P6-ND(NND,K)/HENSU(N3)*P1
1130-1180 REM
1305 N3=N3-1
1505 HENSU2(N3+1)=1-SOWA3
1595 N3=N3+1
1630 REM
1830 REM
```

Program 3. Parts of change from program 1 to the method of least-squares and its variations (the method for the function : f to fit for the frequency distribution : F , COM : common part, B-F as for Table 2)

COM

```
120 REM
990-1040 REM
1130-1180 REM
1470-1500 REM
1830 REM
```

B

```
890 F9=F(K) : G1=GX(K) : F7=2*(F9-G1)
1070 F6=2
1800 F9=F(K)-F1 : LL2=LL2+F9*F9
```

C

```
890 F9=F(K) : G1=GX(K) : F7=F9*F9/G1/G1-1
1070 F6=2*F9*F9/G1/G1/G1
1800 F9=F(K)-F1 : LL2=LL2+F9*F9/F1
```

D

```
890 F9=F(K) : G1=GX(K) : F7=2*F9*(F9-G1)/G1/G1/G1
1070 F6=2*F9*(3*F9-2*G1)/G1/G1/G1/G1
1800 F9=F(K)-F1 : LL2=LL2+F9*F9/F1/F1
```

E

```
885 IF F(K)=0 GOTO 1125
890 F9=F(K) : G1=GX(K) : F7=2*(LOG(F9)-LOG(G1))/G1
1120 NEXT J : NEXT I
1125 NEXT K
1070 F6=2*(1+LOG(F9)-LOG(G1))/G1/G1
1795 IF F(K)=0 GOTO 1820
1800 F9=LOG(F(K)) : G1=LOG(F1) : LL2=LL2+(F9-G1)*(F9-G1)
```

F

```
890 F9=F(K) : G1=GX(K) : F7=2*(LOG(F9+1)-LOG(G1+1))/(G1+1)
1070 F6=2*(1+LOG(F9+1)-LOG(G1+1))/(G1+1)/(G1+1)
1800 F9=LOG(F(K)+1) : G1=LOG(F1+1) : LL2=LL2+(F9-G1)*(F9-G1)
```

Program 4. The common Part of change from program 2 to the method of least-squares and its variations (the method for the provability distribution : g to fit for the distribution of relative frequency : G , B-F as same as program 3)

```
435 FOR I=1 TO NCL : F(I)=F(I)/SOWA : NEXT I
990-1040 REM
```

変数の説明

Correspondence of variables

NND	: number of normal distributions
MCM	: minimum class mark
NCL	: number of classes
CWD	: class width
NIT	: number of iterations
LAMBDA	: λ
NU	: ν
F (K)	: f'
X (K)	: x
SOWA	: T
SOWA2	: S^{OLD}
SOWA3	: S^{NEW}
HENSU (I)	: θ_i^{OLD}
HENSU2 (I)	: θ_i^{NEW}
BIBUN (I)	: $\frac{\partial f}{\partial \theta_i}$
KEISU (I)	: H
TEISU (I)	: g
ZOBUN (I)	: $\Delta \theta_i$
ND (I, K)	: $K_i N(\mu_i, \sigma_i, x)$
LL1	: Y^{OLD}
LL2	: Y^{NEW}
GX (K)	: f

II. 誤差推定

1 2次近似

一般的に目的関数の値は比較的急速に収束するが、各パラメータの収束は遅い。これは解の近傍ではパラメータの偏導関数の値が0に近いとみとえられ、各種の最適化法に見られる共通の特性である (RUCKDESCHEL 1982)。このためパラメータの誤差推定を行なう必要がある。ここではヘシアン行列 H の逆行列の要素および固有値・固有ベクトルを調べて誤差推定を行なった。

目的関数 Y を解の近傍で TAYLOR 展開して2次項までとると

$$\Delta Y = {}^t \left(\frac{\partial Y}{\partial \theta} \right) \Delta \theta + \frac{1}{2} {}^t \Delta \theta H \Delta \theta$$

(${}^t A$ は A の転置行列)

となる。解の近傍では $\frac{\partial Y}{\partial \theta} = 0$ なので

$$\Delta Y = \frac{1}{2} {}^t \Delta \theta H \Delta \theta \tag{29}$$

となる。今回のプログラムはすべて極小値を求めているので $\Delta Y > 0$ となっている。以下の誤差推定はすべて29の近似式に従って行なう。

$$\ell^2 = \Delta \theta_1^2 + \dots + \Delta \theta_n^2 = {}^t \Delta \theta \Delta \theta \tag{30}$$

として各パラメータの動きを調べる。LAGRANGE の乗数法より

$$Q = \ell^2 - \lambda ({}^t \Delta \theta H \Delta \theta - 2 \Delta Y) \tag{31}$$

とおく、 $\Delta Y = \text{定数}$ とすると31式は楕円面となっている (図5参照)。これより




図5 解の近傍

Fig. 5. Illustration of the neighborhood of the solution.

Y : object function θ : parameter

Y_0, θ_0 : solution

e_1, e_2 : eigenvector of H^{-1}

$$\frac{\partial Q}{\partial \Delta \theta} = 2(\Delta \theta - \lambda H \Delta \theta) = 0 \quad \text{より}$$

$$H \Delta \theta = \frac{1}{\lambda} \Delta \theta \quad \text{---32}$$

$$H^{-1} \Delta \theta = \lambda \Delta \theta \quad \text{---33}$$

となる。したがって H^{-1} の固有値と固有ベクトルを求めればよい。

固有値と固有ベクトルの一般的説明を以下に述べる。Aを正則な行列とする。

$$A\mathbf{x}_i = \lambda_i \mathbf{x}_i \quad \text{--- 34}$$

のとき、 λ_i と \mathbf{x}_i をAの固有値・固有ベクトルと言う。34式を満たすベクトルのうち正規直交ベクトルを

$${}^t\mathbf{e}_i \mathbf{e}_j = \begin{cases} 0 & (i \neq j) \\ 1 & (i = j) \end{cases}$$

とする。

$$\begin{aligned} P &= (\mathbf{e}_1, \dots, \mathbf{e}_n) && \text{とおくと} \\ {}^t P P &= I, \quad {}^t P = P^{-1} && \text{--- 35} \end{aligned}$$

Pは直交行列である。このとき

$${}^t P A P = \begin{pmatrix} \lambda_1 & & \\ & \ddots & \\ & & \lambda_n \end{pmatrix} \quad \text{--- 36}$$

となる(固有値分解)。また

$$\begin{aligned} A &= P \begin{pmatrix} \lambda_1 & & \\ & \ddots & \\ & & \lambda_n \end{pmatrix} {}^t P \\ &= P \left\{ \begin{pmatrix} \lambda_1 & & \\ & & \\ & & \end{pmatrix} + \dots + \begin{pmatrix} & & \\ & & \\ & & \lambda_n \end{pmatrix} \right\} {}^t P \\ &= \lambda_1 \mathbf{e}_1 {}^t \mathbf{e}_1 + \dots + \lambda_n \mathbf{e}_n {}^t \mathbf{e}_n \end{aligned} \quad \text{--- 37}$$

このとき

$$I = \mathbf{e}_1 {}^t \mathbf{e}_1 + \dots + \mathbf{e}_n {}^t \mathbf{e}_n$$

である(スペクトル分解)。37式よりAの対角要素の和(Tr A)は

$$\begin{aligned} \text{Tr } A &= \lambda_1 \text{Tr}(\mathbf{e}_1 {}^t \mathbf{e}_1) + \dots + \lambda_n \text{Tr}(\mathbf{e}_n {}^t \mathbf{e}_n) \\ &= \lambda_1 {}^t \mathbf{e}_1 \mathbf{e}_1 + \dots + \lambda_n {}^t \mathbf{e}_n \mathbf{e}_n \\ &= \lambda_1 + \dots + \lambda_n \end{aligned} \quad \text{--- 38}$$

となっている。

33式よりH⁻¹の固有値と正規直交化した固有ベクトルを λ_i, \mathbf{e}_i とおくと

$$\begin{aligned} H \mathbf{e}_i &= \frac{1}{\lambda_i} \mathbf{e}_i \quad \text{だから } \Delta \theta = \mathbf{e}_i \text{ のとき} \\ 2 \Delta Y &= {}^t \mathbf{e}_i H \mathbf{e}_i = \frac{1}{\lambda_i} {}^t \mathbf{e}_i \mathbf{e}_i = \frac{1}{\lambda_i} \end{aligned} \quad \text{--- 39}$$

となる。したがって

$$\mathbf{x}_i = k \mathbf{e}_i \quad \text{--- 40}$$

とおけば $\Delta \theta = \mathbf{x}_i$ のとき

$$\Delta Y = \frac{1}{2} \frac{k^2}{\lambda_i} \quad \text{--- 41}$$

となる。この近似式はkが小さい場合かなりよくあてはまる。

次に29式より

$$(\Delta \theta {}^t \Delta \theta) H (\Delta \theta {}^t \Delta \theta) = 2 \Delta Y (\Delta \theta {}^t \Delta \theta) \quad \text{--- 42}$$

となる。(Δθ^tΔθ)はrank 1で正則ではないので、2ΔY=1のときHは(Δθ^tΔθ)の一般逆行列となっている。ここで共分散行列

$$V = \Sigma_{\theta} = \langle \Delta \theta {}^t \Delta \theta \rangle \quad \text{--- 43}$$

($\langle \rangle$: 期待値)

を考える, H^{-1} の固有ベクトルを次のように定める,

$$\mathbf{x}_i = \sqrt{\lambda_i} \mathbf{e}_i \quad \text{したがって}$$

$${}^t \mathbf{x}_i \mathbf{x}_j = \begin{cases} 0 & (i \neq j) \\ \lambda_i & (i = j) \end{cases} \quad \text{---44}$$

このとき

$$\Delta \theta = a_1 \mathbf{x}_1 + \dots + a_n \mathbf{x}_n \quad \text{---45}$$

と表わせる, ④5式を②9式に代入すると

$$a_1^2 + \dots + a_n^2 = 2 \Delta Y \quad \text{---46}$$

④6式の制限下で④3式を考えると

$$\langle a_i a_j \rangle = \begin{cases} 0 & (i \neq j) \\ \frac{2 \Delta Y}{n} & (i = j) \end{cases} \quad \text{---47}$$

となるので

$$\begin{aligned} \Sigma_{\theta} &= \frac{2 \Delta Y}{n} (\mathbf{x}_1 {}^t \mathbf{x}_1 + \dots + \mathbf{x}_n {}^t \mathbf{x}_n) \\ &= \frac{2 \Delta Y}{n} (\lambda_1 \mathbf{e}_1 {}^t \mathbf{e}_1 + \dots + \lambda_n \mathbf{e}_n {}^t \mathbf{e}_n) \end{aligned}$$

したがって

$$\Sigma_{\theta} = \frac{2 \Delta Y}{n} H^{-1} \quad \text{---48}$$

となる, つまり

$$V \sim H^{-1} \quad \text{---49}$$

したがって H^{-1} の各要素は $\Delta \theta$ の共分散に対応している,

$$H^{-1} = (h_{ij}) \quad \text{として}$$

$$R = \left(\sqrt{\frac{h_{ij}}{h_{ii} h_{jj}}} \right) \quad \text{---50}$$

を計算すれば R は相関行列となる,

以上より, まず H^{-1} を計算して相関行列を求めれば各パラメータの変動の傾向を知ることができる, 次に H^{-1} の固有値と固有ベクトルを求めれば④1式より ΔY の近似値を得ることができ, 非線型性の程度や解の近傍での状態をより具体的に調べることができる,

2 検定

最尤法では尤度比検定を行なうことができる, 帰無仮説: H_0 を以下のように定義する,

$$H_0 : \theta \in \omega \quad (\omega \subset \Omega) \quad \text{---51}$$

Ω : 母数空間

このとき尤度比: λ は

$$\lambda = \frac{\max_{\theta \in \omega} L(\theta)}{\max_{\theta \in \Omega} L(\theta)} \quad \text{---52}$$

と定義される, $n \rightarrow \infty$ のとき $-2 \log \lambda$ は自由度 $(k-s)$ の χ^2 分布に従う, k は対立仮説での独立な母数の個数であり, s は帰無仮説での独立な母数の個数である, 具体的には

$$H_0 : \theta_1 = \theta_{10}, \dots, \theta_r = \theta_{r0}$$

($\theta_{10} \sim \theta_{r0}$ はある定まった値)

とすると $-2 \log \lambda$ の値は自由度 r の χ^2 分布に従う,

$$\log \lambda = \max_{\theta \in \omega} \log L(\theta) - \max_{\theta \in \Omega} \log L(\theta)$$

であるから $\log L$ の値の差で容易に検定できる。

最小二乗法では線型の場合

$$\frac{\Delta Y}{Y} = \frac{p}{m-p} F(p, m-p, 1-\alpha) \quad \text{---(53)}$$

$$\begin{cases} m : \text{測定数 (階級の数)} \\ p : \text{パラメータ数} \\ \alpha : \text{確率水準} \end{cases}$$

となるので (DRAPER and SMITH 1968), 非線型の場合に ΔY を考える目安になる。しかし、この場合には誤差が正規分布に従うという仮定に無理があるので厳密な検定ではない。

3 計算手順

プログラムでは KEISU (I, J) に H の値, TEISU (I) に g の値がはいっている。したがって初期値として解の値を入力し, 集計の終わった時点 (プログラムでは80行の後) でストップさせて, KEISU (I, J) と TEISU (I) を出力させればよい。解に十分近ければ TEISU (I) の値は 0 に近い。このとき H^{-1} を計算すれば(50)式より相関行列 R を得ることができる。

ただし最尤法の場合(10)式に従った方法では H が正則ではない (rank が1つ少ない) ため H^{-1} が存在しない。つまり解の近傍で誤差曲面が楕円面ではなく, 楕円面が一方方向に無限に引き延ばされた形になっている。そこで(9)式に従った方法 (プログラム 2) で計算し, P_n をのぞく $(3n-1)$ 個のパラメータについて解析した。 P_n は以下の式に従って計算した。

(9)式より

$$\Delta P_n = -\sum_{i=1}^{n-1} \Delta P_i \quad \text{---(54)}$$

である。しかし, パラメータはすべて(23)式によってスケーリングされているので実際には

$$\Delta P_n = -\frac{\sum_{i=1}^{n-1} P_i \Delta P_i}{P_n} \quad \text{---(55)}$$

で計算しなくてはならない。このとき P_n の共分散は同様に

$$\langle P_n \Delta \theta \rangle = \frac{-\sum_{i=1}^{n-1} P_i \langle \Delta P_i \Delta \theta \rangle}{P_n} \quad \text{---(56)}$$

で求める。

なお逆行列の計算には赤嶺 (1982a) のプログラムを使用し, 固有値・固有ベクトルの計算は玄・井田 (1983) のプログラム (HOUSEHOLDER 変換, 二分法, WIELANDT の逆反復法) で求めた。ただし小さい方の固有値・固有ベクトルは精度が悪くて求まらないので, H の大きい方の固有値・固有ベクトルとして求めた (32, 33式)。

III. 応 用 例

田中 (1956) で用いられたキダイのデータを用いて各手法の比較および誤差推定を行なった。データを表 1 に示す。このデータは以下の特徴を備えている。

- 1) 実際に現場で得られたデータであり, 標本数が大きい。
- 2) 小さい方の 2 群 (第 1, 2 群) は間に極小値が存在して明確に区別できるが, 大きい方の 3 群 (第 3 ~ 5 群) は間に極小値がなく連続的に減少していて mode がはっきりしない。
- 3) 異常に大きな 5 個体 (F. L. 31 ~ 36 cm) が存在する。

いずれにしても現場で得られるデータとしてはほぼ理想的なデータと言えよう。

表1 キダイ体長組成データ

Table 1. Data of the fork length distribution of porgy.

F. L. a)	O. F. b)	F. L.	O. F.	F. L.	O. F.
7 - 8	7	17 - 18	448	27 - 28	114
- 9	79	- 19	512	- 29	64
10	509	20	719	- 30	22
- 11	2240	- 21	673	- 31	0
- 12	2341	- 22	445	- 32	2
- 13	623	- 23	341	- 33	2
- 14	476	- 24	310	- 34	0
- 15	1230	- 25	228	- 35	0
- 16	1439	- 26	168	- 36	1
- 17	921	- 27	140		
				Total	14054

a) fork length (cm)

b) observed frequency

表2-a 計算結果

Table 2-a. Solutions of each object function (data for observed frequency : F. L. 7-36, see Table 1).

object function a)		A	B	C	
Solution		A1	B1	C1	IV b)
σ	1	.87	.82	.90	1.00
	2	1.13	1.16	1.12	1.00
	3	1.52	1.48	1.38	1.50
	4	.96	1.10	.62	1.50
	5	2.00	1.70	3.16	1.50
μ	1	11.0	11.0	11.0	11.0
	2	15.3	15.3	15.3	15.5
	3	19.8	19.8	19.7	20.0
	4	23.3	23.4	23.2	24.0
	5	25.6	26.3	23.2	27.0
P (%)	1	41.1	40.5	41.3	
	2	30.3	30.9	30.0	
	3	19.7	19.4	14.7	
	4	3.2	4.8	.8	
	5	5.7	4.3	13.3	
K	1	5771	5675	5826	5000
	2	4262	4331	4232	4000
	3	2770	2721	2074	3000
	4	454	671	108	1000
	5	797	605	1876	500
Total		14054	14003	14116	13500

1 計算結果

表1のデータを入力した計算結果と初期値を表2-aに示す。最小二乗法系手法にはプログラム3を使用した。ただし②⑥～②⑧式の方法では収束が不安定で時間もかかり、マイナスの値が出たり σ が異常に大きくなったりして明らかに不適当な解に収束したため省略した。最尤法の解を見るとF. L. 31~36cmの大きな5個体のため第5群が大きくなりすぎているように思える。そこでこの5個体をのぞいたデータについて計算を行ない結果を表2-bに示す。この場合すべての式で無事収束した。またX-Yプロッターによる作図の一部を図6に示す。使用した小型計算機はNECのPC8801, 9801, 9801Fで9801Fの場合10分程度(反復13回)で計算を終了した。

表2-bよりまず気のつく点は②⑥式と②⑦式の解が非常に近い点である。これは両式とも F/f の値が1に近づくようになっているからである。表2-bの値をそれぞれの目的関数に代入して真の解との差を比較したのが表3である。これより最尤法に最も近い解に収束したのは χ^2 最小化法であり次いで最小二乗法であることがわかる。他の3手法はいずれも不適当と

表2-b 計算結果

Table 2-b. Solutions of each object function (data for observed frequency : F. L. 7-31, see Table 1).

object function a)		A	B	C	D	E	F
Solution		A2	B2	C2	D2	E2	F2
σ	1	.87	.82	.90	1.03	1.03	1.03
	2	1.14	1.16	1.12	1.07	1.07	1.04
	3	1.43	1.48	1.48	1.54	1.54	1.82
	4	1.55	1.10	1.37	1.29	1.29	1.37
	5	1.19	1.69	1.32	1.46	1.46	.82
μ	1	11.0	11.0	11.0	11.0	11.0	11.0
	2	15.3	15.3	15.3	15.3	15.3	15.3
	3	19.7	19.8	19.8	19.8	19.8	19.9
	4	23.5	23.4	23.5	23.5	23.5	24.5
	5	27.2	26.3	27.0	26.8	26.8	27.8
P (%)	1	41.1	40.5	41.3	40.7	38.7	38.8
	2	30.5	30.9	30.2	29.9	30.9	29.9
	3	18.3	19.4	19.0	20.0	20.7	23.7
	4	7.7	4.8	6.6	6.0	6.2	5.9
	5	2.4	4.3	2.9	3.4	3.5	1.6
K	1	5769	5676	5827	5631	5173	5202
	2	4289	4331	4252	4144	4132	4009
	3	2566	2721	2675	2765	2762	3179
	4	1088	673	932	834	833	788
	5	337	603	409	465	466	221
Total		14049	14004	14095	13839	13366	13399

a) A : $F \log \frac{f}{S}$ B : $(F-f)^2$ C : $\frac{(F-f)^2}{f}$ D : $\frac{(F-f)^2}{f^2}$

E : $(\log F - \log f)^2$ F : $\{\log (F+1) - \log (f+1)\}^2$

b) initial value

表2-c 計算結果

Table 2 c. Solutions of each object function (data for observed frequency : F .L. 7 31, see Table 1).

object function c)		B'	C'	D'	E'	F'
Solution		B3	C3	D3	E3	F3
σ	1	.82	.90	1.03	1.03	.82
	2	1.16	1.12	1.06	1.05	1.16
	3	1.49	1.48	1.53	1.53	1.48
	4	1.09	1.37	1.29	1.27	1.10
	5	1.75	1.32	1.46	1.47	1.73
μ	1	11.0	11.0	11.0	11.0	11.0
	2	15.3	15.3	15.3	15.3	15.3
	3	19.8	19.8	19.8	19.8	19.8
	4	23.4	23.5	23.5	23.5	23.3
	5	26.3	26.8	26.8	26.7	26.3
P (%)	1	40.5	41.3	40.9	39.3	40.5
	2	30.9	30.2	30.0	31.1	30.9
	3	19.5	19.0	19.9	20.3	19.4
	4	4.7	6.6	6.0	5.9	4.7
	5	4.5	2.9	3.3	3.4	4.5

$$c) \quad B' : (G-g)^2 \quad C' : \frac{(G-g)^2}{g} \quad D' : \frac{(G-g)^2}{g^2}$$

$$E' : (\log G - \log g)^2 \quad F' : \{\log(G+1) - \log(g+1)\}^2$$


判断できる。これは度数の大きい部分での差が大きくなるためと考えられる。

また F.L. 31~36cmの大きな5個体を含めたデータから得られた解(表2-a)を最尤法に代入してみると表3の下段のようになった。最尤法, 最小二乗法, χ^2 最小化法の順であり, 特に最小二乗法はとびはなれた少数個体のデータについては安定していることがわかる。このようなデータをそのまま使用するには χ^2 最小化法よりもむしろ最小二乗法の方が適切であると判断できる。

次に最小二乗法系手法にプログラム4を使用した計算結果を表2-cに示す。B3~E3の値はB2~E2の値とほとんど同じである。これは表2-bでSとTの値がかなり近い値を示していることから当然の結果と考えられる。しかし, F3の値はF2の値とは全く異なる値に収束しB3と一致している。これは $F+1 \approx F$ であるのに対し, $G+1 \approx 1$ となっているため $\log(1+G) \approx G$ が成立して, ⑳式が㉑式とほとんど同じになっているためと考えられる。個体数を扱う場合, 対数変換の際に1を加える手法はよく使用されるので, ㉑式はそれなりに有効であるが, FのかわりにGを用いる場合には㉑式はほとんど意味をもたない。

なお, プログラム3では $\lambda=10000$ ですべて収束したが, プログラム4ではB3から順に, $\lambda=0.0001, 1, 10000, 10000, 1$ として与えた。MARQUARDT法では目的関数やデータごとに適切な λ の初期値を入れる必要がある。 λ が大きすぎると収束が遅く, 小さすぎると発散したりエラーを生じる。

KIDAI



KIDAI





図6 キダイの体長組成と計算結果
Fig. 6. Fork length frequency curve of porgy drawn by X-Y plotter.
Histogram : observed frequency
Lines : solutions of each object function (see Table 2)

KIDAI



KIDAI




図6 (つづき)

Fig. 6. (continued)

表3 各手法の解の比較

Table 3. Comparison of solutions of each object function
(data for observed frequency : F. L. 7-31, see Table 1.
data for solutions and object functions : see Table 2).

solution object function		ΔY a)						IV	Y_0 b)
		A2	B2	C2	D2	E2	F2		
A		1	24	5	108	132	141	271	37606
B		20609	7646	43772	353655	578262	571352	752675	6246
C		18.66	176.48	1.39	189.01	294.87	314.56	572.2	91.0
D		59.04	483.67	17.71	0.022	0.040	0.697	1.200	1.318
E		4.783	10.676	2.608	0.087	0.020	0.456	1.043	0.349
F		3.308	7.627	3.236	2.837	2.774	0.046	5.622	0.878
solution object function		A1	B1	C1					
		A	9	24	34				
									37606

- a) $\Delta Y = Y' - Y_0$
- b) minnum solution of each object function

2 誤差推定

最尤法の解について誤差推定を行なった。表4に g と H の値、表5に H^{-1} ($\sim V$) と R の値を示す。 H^{-1} の対角要素を比較すると以下のことがわかる。

- 1) 全体的にみて μ は σ , P よりも安定である。
- 2) 山が明確に区別できる第1, 2群では σ , P も安定している。

R の各要素を比較してみると以下のことがわかる。

- 3) 強い正の相関がみられるパラメータは $(\sigma_3, \sigma_5, P_3, P_5, \mu_3, \mu_4)$ と (σ_4, P_4, μ_5) の2つのグループに分けることができ、この2つのグループ間には強い負の相関がある。
- 4) 各群ごとにみると σ と P は正の相関が認められる。

まとめて考察すると、各群の中心付近は隣りの群の影響が小さく、 μ の値と $f(\mu)$ の値は安定している。したがって σ_i が大きくなると P_i も大きくなるので各群ごとの σ と P には正の相関が認められる。また山が連続している部分では互いに隣り合った群と領分をとり合うような状態となり、 σ と P は隣りの群とは強い負の相関が生じる。 μ については両端が制限されているため μ_3, μ_5 は逆方向に向かう状態にある。

次に H^{-1} の固有値と固有ベクトルを表6に示す。 $\lambda_1/(\lambda_1 + \dots + \lambda_{14}) = 0.935$ から e_1 で解の近傍の状態をほとんど説明できることがわかる。 e_1 によって V の値はほとんど決定されており(37式)、 e_1 の各要素は前段の議論通りの値を示している。

次に表7に(41式)の値と実際の値を示す。(41式)によってかなり正確に近似できており、解の近傍の状態がよくわかる。ただし k が大きくなると非線型性が強くなり(41式)の値からずれてくるので、この辺の状態を正確に把握するには、たとえば $e_1 - e_2$ 平面上で実際の ΔY の値の等高線を描く等の手法が有効であろう。

表5 H^{-1} と R の値
 Table 5. Elements of H^{-1} and R (H : see Table 4).
 $H^{-1} \times 10^6$

	σ					μ					P				
	1	2	3	4	5	1	2	3	4	5	1	2	3	4	5
σ	126	-71	110	-165	38	3	0	2	11	-5	17	48	124	-237	141
	-2	661	-1515	2383	-568	-7	27	-41	156	74	51	389	-1713	3398	-2008
	1	-6	9642	-22380	6279	12	-126	566	1198	-800	85	-1429	11880	-29332	21013
	0	3	-7	93794	-35390	-18	212	-2072	2952	4301	-126	2451	-35131	107229	-107921
	0	2	4	-8	21327	4	52	669	523	-2163	29	-607	10972	-39189	50291
μ	1	2	1	-1	0	1	-0	0	1	-1	2	-5	14	-26	7
	0	5	-6	3	-2	-0	4	-4	-13	7	-0	24	142	298	-187
	0	-2	8	-9	6	0	-3	57	87	-84	2	-49	856	-2449	1982
	1	-4	9	-7	3	1	-5	8	197	-76	8	-154	1592	-3877	2223
	-0	2	-5	9	-9	-0	2	-7	-3	252	-4	79	-1375	4762	-5851
P	1	-2	1	-0	0	1	-0	0	1	-0	112	-103	24	-253	28
	-2	7	-7	4	-2	-2	6	-3	-5	2	-5	475	-1699	3359	-2194
	1	-5	9	-9	6	1	-6	9	9	-7	0	-6	16620	-43593	35469
	-1	4	-8	10	-8	-1	4	-9	-8	8	-1	4	-9	127972	119865
	0	-2	6	-9	9	0	-2	7	4	-10	0	-3	7	-9	144603

$R \times 10$

a) $P_3 = -(P_1 \cdot P_1' + \dots + P_4 \cdot P_4') / P_3'$
 $P_i' : A_2$ (see Table 2)

表6 H^{-1} の固有値・固有ベクトル

Table 6. Eigenvalues・eigenvectors of H^{-1} (see Table 5).

		e_1	e_2	e_3	e_{13}	e_{14}
λ_i		.25354	.01200	.00365	.00000178	.00000117
$1/\lambda_i$		3.944	83.335	273.700	561182	853362
$\lambda_i/(\lambda_1+\dots+\lambda_{14})$ (%)		93.47	4.42	1.35	.000656	.000431
σ	1	-.00128	.00735	.01531	.02260	.01242
	2	.01832	-.09328	-.16906	-.00561	-.01045
	3	-.15836	.42763	.48033	-.01064	.00200
	4	.60229	.20053	.54019	-.00051	.00005
	5	-.22731	-.76006	.59435	.00004	-.00000
μ	1	-.00014	.00078	.00157	-.25490	-.96599
	2	.00160	-.00717	-.01137	.91972	.25465
	3	-.01369	.00990	-.01355	.29522	-.04150
	4	-.02061	.07383	.00250	-.03438	.00284
	5	.02730	.06242	-.01076	.00147	-.00009
P	1	-.00112	.00503	.01020	.00950	.00218
	2	.01833	-.08412	-.14317	.00115	-.00352
	3	-.23870	.37168	.26522	-.00616	.00135
	4	.70815	-.18905	-.06346	.00284	-.00023
	5 a)	.68406	1.23705	-.16772	.13968	-.00206

a) $P_5 = -(P_1 \cdot P_1' + \dots + P_4 \cdot P_4') / P_5'$
 $P_i' : A2(\text{see Table 2})$

表7 固有ベクトル上の $\hat{\Delta Y}$ a) と ΔY b) の値

Table 7. $\hat{\Delta Y}$ a) and ΔY b) on the eigenvector(see Table 6).

k	-1.0	-0.1	-0.01	-0.001	0	0.001	0.01	0.1	1.0	Range c)
e_1	$\hat{\Delta Y}$	2.0	0.0	0.0	0.0	0.0	0.0	0.0	2.0	-.02~.03
	ΔY	27.9	0.0	0.0	0.0	0.0	0.0	0.0	13.7	
e_2	$\hat{\Delta Y}$	41.7	0.4	0.0	0.0	0.0	0.0	0.4	41.7	-.03~.02
	ΔY	69.9	0.4	0.0	0.0	0.0	0.0	0.4	---	
e_3	$\hat{\Delta Y}$	136.9	1.4	0.0	0.0	0.0	0.0	1.4	136.9	-.08~.06
	ΔY	1055.6	1.6	0.0	0.0	0.0	0.0	1.1	51.3	
e_{13}	$\hat{\Delta Y}$	280591	2805.9	28.1	0.3	0.0	0.3	28.1	280591	-.01~.006
	ΔY	27539	2625.4	28.2	0.3	0.0	0.3	27.8	2642.6	
e_{14}	$\hat{\Delta Y}$	426681	4266.8	42.7	0.4	0.0	0.4	42.7	426681	.01~.02
	ΔY	18188	4264.2	42.6	0.4	0.0	0.4	42.7	4195.6	

a) $\hat{\Delta Y} = \frac{1}{2} \frac{k^2}{\lambda}$

b) $\Delta Y = \log L(\theta_k) - \log L(\theta_0)$

c) Range = $\{k \mid |\hat{\Delta Y} - \Delta Y| \leq 0.1\}$

3 検 定

自由度14の χ^2 分布の値は 23.68 (5%), 29.14 (1%) である。したがって ΔY の値が 11.84 (5%), 14.57 (1%) より大きくなる解は棄却されることになる。表 3 の値をみると最尤法以外でこの範囲に含まれるのは χ^2 最小化法だけであり最小二乗法も近い。また e_1 において 99% 以内にあるのは $-0.8 \leq k \leq 1.0$ の範囲である。つまりベクトル e_1 上においてこの範囲に含まれる解は 99% の確率で棄却できないことになる。不安定なパラメータではこの範囲が大きくなる。 e_1 上におけるパラメータの変化を表 8 に示す。

表 8 最大固有ベクトル e_1 上のパラメータと $\Delta Y^a)$ の値

Table 8. Values of parameters and $\Delta Y^a)$ on the maximum eigenvector e_1 .

		k	-0.8	-0.2	0	0.3	1.0
σ	1		0.8742	0.8735	0.8733	0.8730	0.8722
	2		1.1197	1.1322	1.1364	1.1427	1.1572
	3		1.6150	1.4788	1.4334	1.3653	1.2064
	4		0.8041	1.3650	1.5519	1.8323	2.4866
	5		1.4007	1.2391	1.1852	1.1044	0.9158
μ	1		11.002	11.001	11.001	11.001	10.999
	2		15.268	15.283	15.288	15.295	15.312
	3		19.946	19.784	19.730	19.649	19.460
	4		23.883	23.593	23.496	23.351	23.012
	5		26.599	27.045	27.193	27.416	27.936
P	1		.41100	.41072	.41063	.41049	.41017
	2		.30078	.30414	.30526	.30694	.31086
	3		.21757	.19140	.18268	.16960	.13907
	4		.03358	.06650	.07747	.09393	.13233
	5		.03707	.02724	.02396	.01904	.00757
K	1		5774	5770	5769	5767	5762
	2		4226	4273	4289	4312	4367
	3		3057	2689	2566	2383	1954
	4		472	934	1088	1320	1859
	5		521	383	337	267	106
$\Delta Y^a)$			14.2	0.2	0.0	0.2	13.7

a) $\Delta Y = \log L(\theta_k) - \log L(\theta_0)$

多次元における区間推定は多次元の領域を 考えねばならず 非線型の場合かなり面倒であるが、この例のように e_1 が大きい場合には e_1 上でのみ推定してもかなり有効な情報が得られるだろう。

また厳密ではないが最小二乗法では $F(15, 9, 0.95) = 3.006$, $F(15, 9, 0.99) = 4.962$ なので $Y_0 = 6246$, ③式より $\Delta Y = 31292$ (95%), $\Delta Y = 51654$ (99%) となる。表 3 より最尤法の解は 95% 信頼区間に含まれており、 χ^2 最小化法の解は 99% 信頼区間に含まれていることがわかる。

IV. 考 察

実際に使用するには最尤法が最良であるが絶対ではない。たとえば $n=1$ のとき σ^2 の最尤推定量は不偏推定量になっていない。最尤法は推定量を構成する1つの手法であると位置づけるのが適当で、最尤法によって得られた推定量が望ましい性質をもっているかどうかは個別に調べるという立場をとらなくてはならない(鷺尾1978)。

高速な計算機が使用可能な場合やデータ数が少ない場合には⑤式にもどって生のデータのまま計算するのがよい。具体的には $F=1$ とにおいて x をすべて配列に読み込めばよい。しかし現状では時間とメモリーを大量にとるため非実用的であろう。

また実際のデータでは漁具効率がきいてきて、特に小さい方のデータは使えない場合がある。このように一部分のデータをカットするロバスト推定的な扱いは最小二乗法系の手法では容易であるが最尤法では困難である。このような場合には最小二乗法系で最も最尤法に近い解を与えた χ^2 最小化法を用いるのがよいであろう。ただし、とびはなれた点等をカットしない場合にはむしろ最小二乗法の方がよい。この場合には $S=T$ となつては不合理なのでプログラム3の方を使用しなくてはならない。

HASSELBLAD (1966) は H^{-1} の対角要素の大きさを検討して、正規分布が接近して μ の差が 2σ より小さくなると誤差が異常に大きくなって推定困難になると指摘している。今回の応用例では μ の差は3.8~4.4, σ は0.87~1.55であり μ の差は 2σ 以上になっている。しかし、山の間に極小値がなく連続的に減少している部分ではパラメータの誤差が大きくなった。解の精度を高くすることよりも区間推定をきちんと押さえることの方が重要である。正規分布が接近しすぎて、このプログラムでは分離できないデータや、誤差が大きくなりすぎるデータについては、年齢形質等の生物学的知見からパラメータを推定するしかないだろう。

正規分布の数が不明な場合の取扱いは注意を必要とする。モデルの正規分布の数を増やせばいくらかでもデータの度数分布に近づけることができる。したがって尤度では判定できない。理論的には赤池の情報量基準(AIC)

$$AIC = -2 \log L + 2k$$

k : パラメータ数

を最小にするように k を定めなくてはならない。しかし、データの度数分布を見ただけで正規分布の数を決定できないようなデータでは、誤差がかなり大きくなりAICの値だけで判断するのはかなり危険である。他の生物学的知見等から判断するのが安全であろう。

このプログラムはさまざまに改良できる。特定のパラメータを固定して他のパラメータだけ動かすプログラム等への改良は⑥式の尤度比検定を行なう場合等に有効であろう。またパラメータの初期値を計算するサブルーチン、CRTディスプレイやX-Yプロッターで作図するサブルーチン等をつけ加えればかなり便利なプログラムになるだろう。またMARQUARDT法は非線型な関数の極値を求める有力な手法なので、最尤法や最小二乗法以外にも広く応用できるだろう。

実際の使用にあたってはプログラム1よりもプログラム2の方が計算量が少なく、誤差推定も行なえるのでプログラム2の方を使用した方がよい。最小二乗法・ χ^2 最小化法をロバスト推定的に使う場合にはプログラム3の方を使用しなくてはならない。山がはっきり区別できるような良いデータであれば、従来の種々の手法も有効であろう。生物学的に正しいパラメータの値を求めるのが本来の目的であるから、データにとらわれて細かい数字にこだわるのは危険である。松宮・田中(1974)が行なったように、とびはなれた点や長く尾をひいた部分は区別

して扱う必要がある。計算機に生のデータをそのままほうり込むのではなく、各群において正規分布しているという仮定から著しくはずれているデータはカットして扱わなければならない。常にデータの異常値等をチェックする態度が最も重要である。

V. ま と め

赤嶺 (1984) の最小二乗法のプログラムを最尤法のプログラムに書きかえて、他の最小二乗法系の手法と比較検討し以下の結論を得た。

- 1) 最尤法が原理的に最良であるが、 χ^2 最小化法・最小二乗法もかなり良い解を与える。
- 2) 最小二乗法系の手法はロバスト推定的扱いが容易である。良いデータについては χ^2 最小化法がすぐれているが、とびはなれた値等のあるデータについては最小二乗法の方がよい。
- 3) 解の信頼区間の推定には H^{-1} の固有値と固有ベクトルを用いるのが有効であり、尤度比検定より信頼区間が推定できる。
- 4) 山の間に極小値が存在して山が明瞭に区別できる部分ではかなり精度の良い解が得られるが、山の間に極小値がない部分では誤差が大きくなる。一般的には μ の値は安定であるが、 $\sigma \cdot P$ の値は不安定である。
- 5) このプログラムで分離できなかつたり、信頼区間が異常に大きくなるような条件の悪いデータについては、他の生物学的知見からパラメータを推定するしかない。

本論を終えるにあたり、X-Yプロッターによる作図および助言をしていただいた日本海区水産研究所資源部加藤史彦主任研究官、南西海区水産研究所内海資源部石岡清英主任研究官に感謝します。

文 献

- 赤嶺達郎 (1982a). 年齢組成が不明な場合のサケの回帰率推定法. 日本研報 (33), 141-145.
 (1982b). Polymodal な度数分布を正規分布へ分解する BASIC プログラム. 日本研報 (33), 163-166.
 (1984). MARQUARDT 法による Polymodal な度数分布を正規分布へ分解する BASIC プログラム. 日本研報 (34), 53-60.
- 安藤洋美・門脇光也 (1980). ピアソンとフィッシャーの喧嘩物語 (II). BASIC 数学, 12月号, 現代数学社, 東京, 32-35.
- DRAPER, N. R. and SMITH, H. (1968). 応用回帰分析. (中村慶一訳). 森北出版株式会社, 東京, 378pp.
- 玄 光男・井田憲一 (1983). パソコン会話型科学技術計算プログラム集. 工学図書株式会社, 東京, 163pp.
- HASSELBLAD, V. (1966). Estimation of Parameters for a mixture of normal distributions. Technometrics, 8 (3), 431-444.
- 松宮義晴・田中昌一 (1974). 体長組成解析によるサンマのいわゆる大型・中型等の検討. 東北水研報 (33), 1-18.
- 松崎憲四郎・松崎加奈恵・小川数也 (1983). 年齢群クラス分けプログラムとその応用. 日本ベントス研究会誌 25, 26-32.
- 中川 徹・小柳義夫 (1982). 最小二乗法による実験データ解析プログラム SALS. 東京大学出版会, 東京, 206pp.
- RUCKDESCHEL, F. R. (1982). 科学計算のための BASIC サブルーチン集2 (下). (長谷川勝也, 石原辰雄訳). 現代数学社, 東京, 245pp.
- 嶋津靖彦 (1979). 体長組成から年齢組成を推定する方法. 昭和54年度漁業資源研究会議, 西日本

底魚部会会議報告, 36-48.

田中昌一 (1956). Polymodal な度数分布の二つの取扱方及びそのキダイ体長組成解析への応用.
東海水研報 (14), 1-13.

戸川隼人 (1971). マトリクスの数値計算. オーム社, 東京, 323pp.

鷲尾泰俊 (1978). 推定と検定. 共立出版株式会社, 東京, 123pp.